
Abstract

In Java bytecode, intra-method subroutines are employed to represent code in “finally”
blocks. The use of such polymorphic subroutines within a method makes bytecode analysis
very difficult. Fortunately, such subroutines can be eliminated through recompilation or
inlining. Inlining is the obvious choice since it does not require changing compilers or
access to the source code. It also allows transformation of legacy bytecode. However, the
combination of nested, non-contiguous subroutines with overlapping exception handlers
poses a difficult challenge. This paper presents an algorithm that successfully solves all
these problems without producing superfluous instructions.

Furthermore, inlining can be combined with bytecode simplification, using abstract byte-
code. We show how this abstration is extended to the full set of instructions and how it
simplifies static and dynamic analysis.

1 Introduction

Java [12] is a popular object-oriented, multi-threaded programming language. Ver-
ification of Java programs has become increasingly important. In general, a pro-
gram written in the Java language is compiled to Javabytecode,a machine-readable
format which can be executed by a Java Virtual Machine (VM) [16]. Prior to execu-
tion, such bytecode must pass a well-formedness test calledbytecode verification,
which should allow a regular Java program to pass but also has to ensure that ma-
licious bytecode, which could circumvent security measures, cannot be executed.
The Java programming language includes methods, which are represented as such
in bytecode. However, bytecode also containssubroutines,functions inside the
scope of a method. A special jump-to-subroutine (jsr) instruction saves the re-
turn address to the stack. A return-from-subroutine (ret) instruction returns from
a subroutine, taking a register containing the return address as an argument. This

This is a preliminary version. The final version will be published in
Electronic Notes in Theoretical Computer Science

URL: www.elsevier.nl/locate/entcs



Artho, Biere

artefact was originally designed to save space for bytecode, but it has three unfor-
tunate effects:

(i) It introduces functionality not directly present in the source language.

(ii) The asymmetry of storing the return address on the stack withjsr and retriev-
ing it from a register (rather than the stack) greatly complicates code analysis.

(iii) A subroutine may read and write local variables that are visible within the
entire method, requiring distinction of different calling contexts.

The second and third effect have been observed by Stärk et al. [21], giving numer-
ous examples that could not be handled by Sun’s bytecode verifier for several years.
The addition of subroutines makes bytecode verification much more complex, as
the verifier has to ensure that noret instruction returns to an incorrect address,
which would compromise Java security [16,21]. Therefore subroutine elimination
is a step towards simplication of bytecode, which can be used in future JVMs, al-
lowing them to dispense with the challenge of verifying subroutines.

Correct elimination of subroutines can be very difficult, particularly with nested
subroutines, as shall be shown in this paper. Furthermore, considering the entire
bytecode instruction set makes for very cumbersome analyzers, because it encom-
passes over 200 instructions, many of which are variants of a base instruction with
its main parameter hard-coded for space optimization [16]. Therefore we intro-
duce a register-based version of abstract bytecode which is derived from [21]. By
introducing registers, we eliminate the problem of not having explicit instruction
arguments, simplifying analysis further.

JNuke is a framework for static and dynamic analysis of Java programs [2,5].
Dynamic analysis,including run-time verification [1] and model checking [13],
has the key advantage of having precise information available, compared to clas-
sical approaches liketheorem proving[10]. At the core of JNuke is its VM. Its
event-based run-time verification API serves as a platform for various run-time al-
gorithms, including detection of high-level data races [4] and stale-value errors [3].

Recently, JNuke has been extended with static analysis [2], which is usually
faster than dynamic analysis but less precise, approximating the set of possible pro-
gram states. “Classical” static analysis uses a graph representation of the program
to calculate a fix point [8]. The goal was to re-use the analysis logics for static and
dynamic analysis. This was achieved by a graph-free data flow analysis [20] where
the structure of static analysis resembles a VM but allows for non-determinism and
uses sets of states rather than single states in its abstract interpretation [2].

Bytecode was the chosen input format because it allows for verification of Java
programs without requiring their source code. Recently, even compilers for other
languages to Java bytecode have been developed, such asjgnat for Ada [7] or
kawa for Scheme [6]. However, bytecode subroutines and its a very large, stack-
based instruction set make static and dynamic analysis difficult. JNuke eliminates
subroutines and simplifies the bytecode instruction set.

Section2 gives an overview of Java compilation and treatment of exception
handlers. The inlining algorithm is given in Section3. Section4 describes conver-

Instruction Description

aload r Pushes a reference or an address from registerr onto the stack.

iload r Pushes an integer from registerr onto the stack.

astore r Removes the top stack element, a reference or address, storing it in registerr.

istore r Removes the top stack element, an integer, storing it in registerr.

goto a Transfers control to the instruction ata.

iinc r j Increments registerr by j.

ifne a Removes integerj from the stack; ifj is not 0, transfers control toa.

jsr a Pushes the successor of the current addr. onto the stack and transfers control toa.

ret r Loads an addressa from registerr and transfers control toa.

athrow Removes referencer from the stack, “throwing“ it as an exception to the caller.

return Returns from the current method, discarding the stack and all local variables.

Table 1
A subset of Java bytecode instructions.

sion to abstract, register-based bytecode. Section5 describes differences between
our work and related projects, and Section6 concludes.

2 Java Compilation with Bytecode Subroutines

2.1 Java Bytecode

Java bytecode [16] is an assembler-like language, consisting of instructions that
can transfer control to another instruction, access local variables and manipulate a
(fixed-height) stack. Each instruction has a unique address orcode index. Table1
describes the instructions referred to in this paper. In this table,r refers to a register
or local variable,j to a (possibly negative) integer value, anda to an address. The
instruction at that addressa will be denoted as code(a), while the reverse of that
function, index(ins) returns the address of an instruction.

The maximal height of the stack is determined at compile time. The type of
instruction argument has to be correct. Register indices must lie within statically
determined bounds. These conditions are ensured by any well-behaved Java com-
piler and have to be verified by the class loader of the Java Virtual Machine (VM)
duringbytecode verification[16], the full scope of which is not discussed here.

2.2 Exception Handlers and Finally Blocks

The Java language containsexceptions,constructs typically used to signal error
conditions. An exception supercedes normal control flow, creates a new exception
object e on the stack and transfers control to anexception handler.The range
within which an exception can be “caught” is specified by atry block. If such
an exceptione occurs at run-time, execution will continue at the corresponding

catch block, if present, which deals with the exceptional program behavior. An
optionalfinally block is executed whether an exception occurs or not, but always
after execution of thetry andcatch blocks. Therefore, the presence of afinally
block creates a dualistic scenario: in one case, an exception occurs, which requires
both thecatch andfinally blocks to be executed. In the absence of an exception,
or if an exception occurs that is not within the type specification of thecatch block,
only thefinally block has to be executed. Because of this, a default exception
handler is required to catch all exceptions that are not caught manually.

In the following text, lower case letters denote single values. Monospaced cap-
ital letters such asC will denote control transfer targets (statically known). Capitals
in italics such asI denote sets or ranges of values. In Java bytecode, an exception
handlerh(t, I ,C) is defined by its typet, rangeI , which is an interval[iα, iω], 1 and
handler code atC. Whenever an exception of typet or its subtypes occurs within
I , control is transferred toC. If several handlers are eligible for rangeI , the first
matching handler is chosen. If, for an instruction indexa, there exists a handlerh
wherea lies within its rangeI , we say thath protects a: protects(h,a)↔ a∈ I(h).

As specified by the Java language [12], a finally block atF always has to be
executed, whether an exception occurs or not. This is achieved by using an un-
specified typetany for a default handlerhd(tany, Id,F). If a catch block is present
in a try/catch/finally construct, the exception handlerh′(t ′, I ′,C′) specified by
thecatch clause takes priority over default handlerhd. Handler code atC′ is only
executed when an exception compatible with typet ′ is thrown. In that case, af-
ter executing thecatch block, agoto instruction is typically used to jump to the
finally block atF. Because this mechanism is a straightforward augmentation of
catching any exception byhd, this causes no new problems for subroutine inlining
and verification. Hencecatch blocks are not discussed further in this paper.

2.3 Finally Blocks and Subroutines

A finally block can be executed in two modes: either an exception terminated
its try block prematurely, or no exception was thrown. The only difference is
therefore the “context” in which the block executes: it possibly has to handle an
exceptione. This lead to the idea of sharing the common code of a finally block.
Thus a Java compiler typically implementsfinally blocks usingsubroutines.2

A subroutineS is a function-like block of code. In this paper,S will refer to the

1 In actual Java class files, handler ranges are defined as[iα, iω[ and donot include the last index
of the interval,iω. This is only an implementation issue. For simplicity, this paper assumes that
handler ranges are converted to reflect the above definition.
2 Sun’s J2SE compilers, version 1.4.2 and later, compilefinally blocks without subroutines.
However, in order to ensure backward compatibility with legacy bytecode, the bytecode verifier still
has to deal with the complexity of allowing for correct subroutines. This underlines the need for
subroutine elimination, as commercial libraries often do not use the latest available compiler but
can still be used in conjunction with programs compiled by them. This paper lays the groundwork
for inlining subroutines in legacy bytecode, allowing bytecode verifiers in future VMs to ignore this
problem.

int m(int i) {
  try {
      i++;
  } finally {
      i--;
  }
  return i;
}

    |    iinc i 1
    |
(h) |    jsr S
    |
    |    goto X
      C: astore e
         jsr S
         aload  e
         athrow
      S: astore r
         iinc i -1
         ret r
      X: iload i
         ireturn

main

goto X

S

C

C: athrow

X

return

Figure 1. A simple finally block, its bytecode and its control flow graph.

entire subroutine whileS denotes the address of the first instruction ofS. A subrou-
tine can be called by a special jump-to-subroutine instructionjsr, which pushes the
successor of the current address onto the stack. The subroutine first has to store that
address in a registerr, from which it is later retrieved by a return-from-subroutine
instructionret. Registerr cannot be used for computations. Java compilers nor-
mally transform the entirefinally block into a subroutine. This subroutine is
called whenever needed: after normal execution of thetry block, after exceptions
have been taken care of withcatch, or when an uncaught exception occurs.

The example in Figure1 illustrates this. RangeR which handlerh(t,R,C) pro-
tects is marked by a vertical line. The handler code atC first stores the exception
reference in a local variablee. It then calls thefinally block atS. After execut-
ing S, the exception reference is loaded from variablee and thrown to the caller
using instructionathrow. If no exception occurs,S is called after thetry block,
before continuing execution atX. Note that the subroutine block is inside the entire
method, requiring agoto instruction to continue execution atX, after thetry block.
In the control flow graph,Scan be treated as a simple block of code which can be
called from the top level of the method (main) or exception handler codeC. In the
first case,S will return (with ret) to instructiongoto X, otherwise to the second
part of the handler ending withathrow.

2.4 Nested Subroutines

The example in Figure2 from [21, Chapter 16] illustrates difficulties when dealing
with subroutines. It contains a nestedfinally block with abreak statement.3 The
compiler transforms this into two exception handlersh1(t1,R1,C1) andh2(t2,R2,C2)

3 The body of the method does not contain any semantically relevant operations for simplicity.
The resulting code, compiled by Sun’s J2SE 1.3 compiler, includes a handler protecting areturn
statement, even though that instruction cannot throw an exception. The handler may come into
effect if thetry block contains additional instructions. Therefore it is preserved in this example.

static void m(boolean b) {
  try {
    return;
  } finally {
    while (b) {
      try {
        return;
      } finally {
        if (b) break;
      }
    }
  }
}

     |     jsr S1
(h1) |
     |     return
       C1: astore e1
           jsr S1
           aload e1
           athrow
       S1: astore r1
           goto W
     |  L: jsr S2
(h2) |
     |     return
       C2: astore e2
           jsr S2
           aload e2
           athrow
       S2: astore r2
           iload b
           ifne X
        Y: ret r2
        W: iload b
           ifne L
        X: ret r1

Figure 2. Breaking out of a subroutine to an enclosing subroutine.

using two subroutinesS1 andS2, where it is possible to return directly to the en-
closing subroutine from the inner subroutine, without executing theret statement
belonging to the inner subroutine. Letteredenotes a register holding a reference to
an exception,r a register holding the return address of a subroutine call.

The corresponding control flow graph in Figure3 is quite complex. Its two
exception handlersh1 andh2 contain onefinally block each. The firstfinally
block contains a while loop with testW and loop bodyL . If the loop test fails,
S1 returns viaX to the successor of its caller. This may be the second instruc-
tion, or code afterC1, which throws exceptione1 after having executedS1. Loop
bodyL contains in innertry/finally statement, compiled into exception handler
h2. Execution ofL results in calling innerfinally block atS2, again prior to the
return statement. This block will testb and break to the outer subroutine, which
is represented by connectionS2 → X. If b was false, the inner subroutine would
return normally using itsret instruction atY. There, control will return to the inner
return statement withinL, which then returns from the method. Bothtry blocks
are also protected by default exception handlers, where the control flow is similar.
The main difference is that an exception will be thrown rather than a value returned.

3 Inlining Java Subroutines

Once all subroutines with their boundaries have been found, they can be inlined.
Inlining usually increases the size of a program only sightly [11] but significantly
reduces the complexity of data flow analysis [11,21].

C1: athrow C2: athrowmain: returnS1: return

main

S1

W

L

X

S2

Y

C1

C2

Figure 3. Control flow graph of nested subroutines.

Instruction (at addresspc) Addresses of possible successors

aload, iload, astore, istore, iinc {pc+1}

goto a {a}

ifne a, jsr a {a, pc+1}

ret, athrow, return {}

Table 2
Potential successors of Java bytecode instructions.

Table2 defines potential successors of all bytecode instructions covered here.
Without loss of generality, it is assumed that instructions are numbered consecu-
tively. Thuspc+ 1 refers to the successor of the current instruction,pc−1 to its
predecessor. Conditional branches (ifne) are treated non-deterministically. The
jsr instruction is modeled to have two successors because control returns topc+1
after execution of the subroutine ata. Certain instructions leave the scope of the
current method (return, athrow) or continue at a special address (ret).

The first instruction of a method is assumed to have code index 0. A code index
i is reachableif there exists a sequence of successors from instruction 0 toi. S is a
subroutine iffi is reachable and code(i) isjsr S. A code indexX is a possiblereturn
from a subroutineif code(S) is astore r, code(X) is ret r, andX must be reachable
fromS on a path that does not use an additionalastore r instruction. A code indexi
belongs to subroutine S, i ∈S, if there exists a possible returnX from that subroutine

S such thatS ≤ i ≤ X. The end of a subroutine S, eos(S), is the highest index
belonging toS. Note that this definition avoids the semantics of nested exception
handler ranges, thus covering each nested subroutine individually. For the purpose
of inlining, we also need the following definitions: Thebodyof a subroutine is the
code which belongs to a subroutineS, where for each code indexi, S < i < eos(S)
holds. This means the body does not include the first instruction,astore r, and the
last instruction,ret r. A subroutineS2 is nestedin S1 if for each code indexi which
belongs toS2, i ∈ S1 holds. From this,S1 < S2 and eos(S1) > eos(S2) follows.
Furthermore, code(S2− 1) must be instructiongoto eos(S2) + 1. A subroutine
S1 is dependent ona (possibly nested) subroutineS2, S1 ≺ S2, if there exists an
instructionjsr S2 which belongs to subroutineS1, whereS2 6= S1. Dependencies
are transitive.

A subroutineS1 which depends onS2 must be inlined afterS2. WhenS1 is
inlined later, the calls toS2 within S1 have already been replaced by the body ofS2.
Other than that, the order in which subroutines are inlined does not matter. During
each inlining step, all calls to one subroutineSare inlined.

3.1 Sufficient and Necessary Well-formedness Conditions

Java bytecode can only be inlined if certain well-formedness conditions hold. A set
of necessary conditions is given by the specification of bytecode verification, which
includes that subroutines must have a single entry point and that return addresses
cannot be generated by means other than ajsr instruction [16]. Beyond these given
conditions, extra conditions have to hold such that a subroutine can be inlined. Note
that it is not possible that programs generated by a Java compiler violate these con-
ditions, except for a minor aspect concerning JDK 1.4, which is described below.
Furthermore, artificially generated, “malicious” bytecode that does not fulfill these
well-formedness criteria will likely be rejected by a bytecode verifier. Bytecode
verification is in essence an undecidable problem, and thus verifiers only allow for
a subset of all possible bytecode programs to pass [16,21].

One extra condition not described here arises from the current artificial size
limit of 65536 bytes per method [16]. Other limitations are structural conditions
that bytecode has to fulfill. Given here is an abridged definition taken from [21]:

Boundary. Each subroutineSmust have an end eos(S).
If subroutineSdoes not have aret statement, then all instances ofjsr S can be
replaced withgoto S, and no inlining is needed.

No recursion. A subroutine cannot call itself.

Correct nesting. Subroutines may not overlap:
@S1,S2 ·S1 < S2 < eos(S1) < eos(S2).

No mutual dependencies.If Si ≺ Sj , there must be no dependencies such that
Sj ≺ Si . Note this property is not affected by nesting.

Exception handler containment. If code C of a handlerh(t,R,C) belongs toS,
then its entire rangeRmust belong toSas well:∀h(t,R,C),S· (C ∈ S→ R⊆ S).

      S: astore r
         ...
         jsr S
         ...
 eos(S): ret r

     S1: astore r1
         ...
     S2: astore r2
         ...
eos(S1): ret r1 
         ...
eos(S2): ret r2

     S1: astore r1
         ...
         jsr S2
         ...
eos(S1): ret r1
         ...
     S2: astore r2
         ...
         jsr S1
         ...
eos(S2): ret r2

No recursion. Correct nesting. No mutual dependencies.

     | : 
     |
   h |   ...
     |
     | : 
         ...
      S: astore r
         ...
      C: 
         ...
 eos(S): ret r

    |  : 
    |
    |    ...
    |
  h | S: astore r
    |
    |    ...
    |
    |  : 
         ...
 eos(S): ret r
         ...
      C: 

          jsr S
          ...
    |   : 
    |
    |     ...
    |
  h |  S: astore r
    |
    |     ...
    |
 eos|(S): ret r
    |
    |     ...
    |
    |   : 

Exception handler Handler range Subroutine containment

containment. containment. in handler range.

Figure 4. Instruction sequences violating well-formedness conditions.

Handler range containment. If any i ∈Rof a handlerh(t,R,C) belongs toS, then
its entire rangeRmust belong toS: ∀h(t,R,C),S· (∃i ∈ R· i ∈ S→ R⊆ S).

Subroutine containment in handler range.
If the entire rangeR of a handlerh(t,R,C) belongs toS, then any instructions
jsr S must be withinR: ∀h(t,R,C),S· (R⊆ S→ (∀i ·code(i) = jsrS→ i ∈R)).

For the last six conditions, Figure4 shows an example violating it. Existing Java
compilers do not violate them except as described in Subsection3.5.

3.2 Control Transfer Targets

When inlining subroutines, the body of a subroutineS replaces each subroutine
call. This part of inlining is trivial, as shown by the example in Figure5. The two
inlined copies ofSwhich replace thejsr instructions are shown inbold face. Dif-
ficulties arise with jump targets, which have to be updated after inlining. Inlining
eliminatesjsr andret instructions; therefore any jumps to these instructions are
no longer valid. Furthermore, there can be jumps inside a subroutine to an enclos-
ing subroutine or the top level of the code, such as shown in Figure6. Therefore,

    |    iinc i 1
    |
(h) |    jsr S
    |
    |    goto X
      C: astore e
         jsr S
         aload  e
         athrow
      S: astore r
         iinc i -1
         ret r
      X: iload i
         ireturn

(h1) |   iinc i 1
        iinc i -1
(h2) |   goto X
      C: astore e
        iinc i -1
         aload  e
         athrow
      X: iload i
         ireturn

Figure 5. Inlining a subroutine.

the inlining algorithm has to update jump targets during several inlining steps and
also to consider copies, for each instance of a subroutine body that gets inlined.

The algorithm uses twocode sets, currentsetB andnewsetB′. During each
inlining step, all instructions inB are moved and possibly duplicated, creating a
new set of instructionsB′ which becomes the inputB for the next inlining step.

Each address inB must map onto anequivalentaddressB′. Each possible exe-
cution (including exceptional behavior) must execute the same sequence of opera-
tions, excludingjsr andret, in B andB′. Code indices inB referring tojsr or ret
instructions must be mapped to equivalent indices inB′. The most straightforward
solution is to update all targets each time after inliningoneinstance of a given sub-
routine. This is certainly correct, but also very inefficient, because it would require
updating targets once for eachjsr instruction rather than each subroutine.

Instead, our algorithm uses amapping M, a relationI × I ′ of code indices map-
ping an indexi ∈ I to a set of indices{i′0, i′1, . . . , i′k} ∈ I ′. This relation, initially
empty, records how an address inB is mapped to one or several addresses inB′.
Each time an instruction at indexi is moved or copied from the current code setB
to the new code setB′ at indexi′, i 7→ i′ is added to the mapping.

Each subroutine is processed inliningall its instances in one step, with the in-
nermost subroutines being inlined first. Instructions not belonging to the subroutine
which is being inlined and which are not ajsr S operation are copied over from
B to B′. Each occurrence ofjsr S is replaced with the body ofS. The key to
handling jumps toinsj , thejsr S instruction itself, and toinsr , theret instruction
in the subroutine, is adding two extra entries toM. The first one isi j → i′0 where
i j = index(insj) and i′0 = M(S), the index where the first instruction of the sub-
routine has been mapped to. The second one isir → i′r whereir = index(insr) and
i′r = M(eos(S)+1), the index of the first instructionafter the inlined subroutine.

In the following discussion, aforward jumpis a jump whose target code index
is greater than the code index of the instruction. Similarly, abackward jumpis a
jump leading to a smaller code index. If bytecode fulfills the correctness criteria
described above, the correctness of the algorithm can be proved as follows:

• A target outsideS is mapped to a single target and therefore trivially correct.

• A target in the body ofS is mapped to several targets in the inlined subroutines
S′, S′′ etc., one for eachjsr S in B. Let the jump instruction inB be at code index
i and the target ata. Giveni′, the index of the jump instruction inB′, thenearest
target in the current mapping has to be chosen which still preserves the fact that
a jump is either a forward or backward jump.
For a forward jump, index mina′ ·(a 7→ a′ ∈ M)∧ (a′ > i′) is the correct index.
This can be shown as follows: Addressa is either outsideS, in which case the
code(a) has not been duplicated and there is only onea′ · a 7→ a′ ∈ M. If a is
insideS, a′ is necessarily the nearest target toi′ in that direction: The code at
indexa has been copied toa′ during the inlining ofS to S′. The first instruction
of the inlined copy ofS′ is at indexS′α and the last instruction is atS′ω. Since
i belongs toS, S′α ≤ i′ ≤ S′ω holds. No other code thanS′ has been copied to
positions inside that interval, andS′α ≤ i′ < a′ ≤ S′ω holds becausea belongs to
Sand the jump is a forward jump. Any other copies of the instructions ata are
either copied to an indexa′′ < S′α, and thereforea′′ < i′, ora′′ > S′ω, and therefore
a′′ > a′. Backward jumps are treated vice versa.

• A jump to ajsr S instruction inB indirectly executes code atS. Mapping it to
theS′α preserves the semantics.

• A jump to the last instruction in a subroutine will return to the successor of its
jsr S instruction. Therefore mapping the code index of theret instruction to
the successor of the last inlined instruction of the body ofSproduces the same
effect in the inlined code. Note that there always exists an instruction following
ajsr instruction [16], such asreturn.

Two of these cases are shown in the second inlining step of Figure6, the inlining
of the subroutines in Figure2. In both inlined instances ofS1, the outer subroutine,
there is a jump toW inside the subroutine and toX, the index of theret instruction of
S1. By inlining S1, both code indices are mapped to two new indices,{W1,W2}, and
{X1,X2}, respectively. The semantics of jumps are preserved as described above.

3.3 Exception Handler Splitting

If a jsr S instruction insj is protected by an exception handlerh(t,R,C), where
R= [rα, rω] doesnot extend to the subroutine itself, then that handler mustnot be
active for the inlined subroutine. A simple example is shown in Figure5, where the
jsr instruction is in the middle of the exception handler range. Therefore, to solve
this problem, the exception handler must besplit into two handlersh1(t,R1,C′)
andh2(t,R2,C′). The new ranges areR1 = [r ′a, rβ] andR2, with r ′α = M(rα) and
rβ = M(index(insj)− 1), the mapped code index of the predecessor of thejsr
instruction. InR2 = [rγ, r ′ω], rγ = M(index(insr)), the mapped code index of the
successor of the last instruction of the inlined subroutine body, andr ′ω = M(rω).

Splitting handlers is necessary to ensure correctness of the inlined program.
There exist cases whereR1 or R2 degenerates to an interval of length zero and can
be removed altogether. Splitting may increase the number of exception handlers

     |     jsr S1
(h1) |
     |     return
       C1: astore e1
           jsr S1
           aload e1
           athrow
       S1: astore r1
           goto W
     |  L: jsr S2
(h2) |
     |     return
       C2: astore e2
           jsr S2
           aload e2
           athrow
       S2: astore r2
           iload b
           ifne X
        Y: ret r2
        W: iload b
           ifne L
        X: ret r1

     |     jsr S1
(h1) |
     |     return
       C1: astore e1
           jsr S1
           aload e1
           athrow
       S1: astore r1
           goto W
       L: iload b
           ifne X
(h2) |     return
       C2: astore e2
          iload b
           ifne X
           aload e2
           athrow
        W: iload b
           ifne L
        X: ret r1

          goto W1
       L1: iload b
           ifne X1
(h21) |    return
      C21: astore e2
           iload b
           ifne X1
           aload e2
           athrow
       W1: iload b
           ifne L1
(h1) | X1: return
       C1: astore e1
          goto W2
       L2: iload b
           ifne X2
(h22) |    return
      C22: astore e2
           iload b
           ifne X2
           aload e2
           athrow
       W2: iload b
           ifne L2
       X2: aload e1
           athrow

Figure 6. Inlining a nested subroutine in two steps

exponentially in the nesting depth of a subroutine. This number is almost never
greater than one, though, and only few exception handlers are affected by splitting.

3.4 Exception Handler Copying

If a subroutineS, but not thejsr S statement, is protected by an exception han-
dler, this protection also has to be ensured for the inlined copy of the subroutine.
Therefore, all exception handlers protecting subroutineShave to becopiedfor each
inlined instance ofS. Figure6 shows a case where inlining the outer subroutineS1

causes the exception handlerh2 inside that subroutine to be duplicated.
Note that this duplication does not occur ifboth the jsr instruction and the

subroutine are protected by the same handler. In this case, the inlined subroutine
is automatically included in the mapped handler range. Copying handlers may
increase the number of handlers exponentially, which is not an issue in practice
because theinnermostsubroutine, corresponding to the innermostfinally block,
is never protected by an exception handler itself, reducing the exponent by one.

3.5 Violation of Well-formedness Conditions in JDK 1.4

The original implementation exhibited problems with some class files compiled
with the JDK 1.4 compiler. The reason were changes in the compiler, designed
to aid the bytecode verifier of the VM. When compiling the program from Figure
1, the resulting instructions are the same, but the exception handlers are different:
The original handler covered three instructions, the initial increment instruction,

Number of calls 1 2 3 4 5 6 – 10 11 – 20 28

Number of subroutines 1 783 173 23 9 8 3 1

Table 3
Number of calls per subroutine, determining how often its code is inlined.

thejsr, and thegoto which jumps to the end of the program. The handler from
the 1.4 compiler omits thegoto. This does not change the semantics of the code
because thegoto instruction cannot raise an exception.

However, asecondhandlerh is installed by the newer compiler, which covers
the first two instructions of the exception handler code (at labelC), astore eand the
second instance ofjsr S. The situation is exacerbated by the fact thath is recursive;
the handler code has the same address as the first instruction protected by it. This
could (theoretically) produce an endless loop of exceptions. The result of inlining
h is a handler covering only theastore instruction (since the inlined subroutine
is outside the handler range). Fortunately, theastore instruction cannot throw an
exception, so no changes are needed in the VM to avoid a potentially endless loop.

Newer JDK compilers (1.4.2 and later) generate subroutines in-place. The re-
sult is identical to inlined code from JDK 1.4, including spurious handlerh.

3.6 Costs of Inlining

Inlining subroutines increases code size only slightly. Subroutines are rare. In Sun’s
Java run-time libraries (version 1.4.1), out of all 64994 methods from 7282 classes
(ignoring 980 interfaces), only 966 methods (1.5 %) use 1001 subroutines. None
of them are nested. Table3 shows that subroutines are usually called two to three
times each, with a few exceptions where a subroutine is used more often.

The histogram to the left in Figure7 shows that most subroutines measure only
between 8 and 12 bytes; 626 subroutines were 9 bytes large, hence that entry is
off the scale. No subroutine was larger than 37 bytes. Inlining usually results in a
modest code growth of less than 10 bytes. This is shown by the histogram to the
right where entries with an even and odd number of bytes are summarized in one
bucket. Entries off the scale are values 0 (64041 methods, including those without
subroutines) and 2, representing 571 methods where code size increased by 2 or
3 bytes. 10 methods grew by more than 60 bytes, 186 bytes being the worst case.
Inlining all subroutines of JRE 1.4.1 would result in a code growth of 5998 bytes,
which is negligible compared to the entire run-time library, measuring 25 MB.

4 Abstract, Register-based Bytecode

Java bytecode contains 201 instructions [16], many of which are variants of the
same type. For instance, 25 instructions load a register on the stack. Variants in-
clude several instructions for each data type, one generic variant (e.g.iload r) and
short variants likeaload_0, wherer is hard-coded. A reduction of the instruction

Figure 7. Sizes of subroutines and size increase after inlining.

set is an obvious simplification. We use abstract bytecode [21] as the reduced for-
mat, where argument types and hard-coded indices are folded into the parametrized
version of a generic instruction. For instance,aload_0 becomesLoad “ref” 0. This
reduction is independent of bytecode inlining. The previous section described in-
lining using normal bytecode to allow for stand-alone inlining algorithms.

Instructions not implemented in [21] include arithmetic instructions, of which
implementation is straightforward. Unsupported instructions areswitch (for con-
trol flow), monitorenter andmonitorexit (for multi-threading), and thewide
instruction that modifies the parameter size of the subsequent instruction. The first
three instructions have to be implemented according to the standard bytecode se-
mantics [16] while thewide instruction is an artefact of the fact that Java bytecode
was initially targetted to embedded systems with little memory for instructions. In
our implementation [5] of the abstract bytecode instruction set, we extended the
size of any instruction parameters to four bytes and thus could eliminate the wide
instruction trivially, by converting all instruction arguments to a four-byte format.

Abstract bytecode only has 31 instructions, which is already a great simplifi-
cation of the original instruction set. However, the usage of a (fixed-size) stack
makes data flow analysis needlessly difficult, since the exact stack height at each
index, though known at compile-time, has to be computed first after loading a class
file. This computation is normally part of bytecode verification in the class loader.
Furthermore, the treatment of stack and local variables (registers) results in pairs
of instructions that essentially perform the same task:load pops the top element
from the stack whilestore pushes a register onto the stack. Finally, 64-bit values
are treated as a single stack element, but as a pair of local variables. This creates
a need for case distinctions for many instructions [16]. The specification requires
that the second slot of the local variables holding a 64-bit value is never used, and
that the stack semantics are preserved when pushing a 64-bit value onto it.

Because the height of the stack is known for each instruction, we converted the
stack-based format of abstract bytecode to an explicit representation where each
stack element is converted to a register. When using registers, stack elements and
local variables can be treated uniformly, mergingLoad andStore into a Get in-
struction, and eliminating more instructions such asPop, Swap, or Dup. Of all
conversions, converting theDup instruction was the only non-trivial one and actu-

Bytecode variant Java [16] Abstract [21] Register-based

Instruction set size 201 31 25

Variants (type/index) per instruction up to 25 1 1

Bytecode subroutines yes yes no

Wide instructions yes not impl. eliminated

Special treatment of 64-bit values yes not impl. eliminated

Register location implicit implicit explicit

Table 4
The benefits of register-based bytecode.

ally proved to be quite difficult. Some variants of this instruction do not only copy
the top element(s) of the stack, but insert it “further down”, below the top element.
There exist six variants ofDup instructions, and the treatment of data flow requires
up to four case distinctions per instruction variant, due to 64-bit values [16]. We
convert allDup instructions into an equivalent series ofGet instructions. This un-
fortunately introduces sequences of instructions that corresponds to only one orig-
inal instruction, which makes further treatment slightly more complex; but it still
eliminates the case distinctions for 64-bit values, which is the greater overhead.

This conversion to register-based bytecode reduces the size of the final instruc-
tion set to a mere 25 instructions. The remaining instructions are (refer to [16,21]
for their semantics):ALoad, AStore, ArrayLength, Athrow, Checkcast, Cond,
Const, Get, GetField, GetStatic, Goto, Inc, Instanceof, InvokeSpecial,
InvokeStatic, InvokeVirtual, MonitorEnter, MonitorExit, New, NewArray,
Prim, PutField, PutStatic, Return, Switch. This instruction set was used in
JNuke and has been tested in over 1,000 unit and system tests using static analysis,
run-time verification, and software model checking [2,3,5].

5 Related Work

Previous work has investigated difficulties in analyzing Java bytecode arising from
its large instruction set and subroutines. Inlining bytecode subroutines has been
investigated in the context of just-in-time-compilation [15] or as a preprocessing
stage for theorem proving [11]. The latter paper also describes an alternative to
code duplication for inlining: by storing a small unique integer for each subroutine
call instruction in an extra register, subroutines can be emulated without using ajsr
instruction. However, the size gain by this strategy would be small, and bytecode
verifiers would again have to ensure that the content of this extra register is never
overwritten inside the subroutine, which would leave one of the major problems in
bytecode verification unsolved. Therefore this direction was never pursued further.

Challenges in code analysis similar to those described here occur fordecom-
pilation, where the structure of subroutines must be discovered to determine the

correct scope oftry/catch/finally blocks. The Dava decompiler, which is part
of the Soot framework, analyzes these structures in order to obtain an output that
correctly matches the original source program [19]. Soot also eliminatesjsr in-
structions through inlining [22]. However, no algorithm is given. Details on how
to handle nested subroutines are missing.

As a part of work onµJava [14], another project also performs a kind of sub-
routine inlining called subroutineexpansion[24]. The main difference is that the
expanded code still containsjsr instructions, making it easier to ensure correct-
ness of the inlined code, but still posing a certain burden on the bytecode verifier
that our work eliminates. The inlining algorithm differs in several points. First,
it uses “complex addresses” to track code duplication. Second, it does not inline
subroutines in the order of their nesting. This has two side-effects: treatment of
nested subroutines creates a very complex special case, and the expanded code
may be larger than necessary [24]. Our algorithm uses a simple mapping instead of
complex addresses, which, together with inlining subroutines in the order in which
they are nested, greatly simplifies the adjustment of branch targets and exception
handler ranges. Furthermore, with nesting taken care of by inlining subroutines in
nesting order, no special treatment of nested subroutines is necessary in the inner
loop that performs the actual inlining.

Instruction set reduction on Java bytecode has been performed in other projects
in several ways. The Carmel [17] and Jasmin [18] bytecode instruction sets both
use a reduced instruction set similar to abstract bytecode [21]. The Bytecode Engi-
neering Library (BCEL) does not directly reduce the instruction set but features an
object-oriented representation of bytecode instructions where super classes com-
bine related instructions [9]. The project most similar to ours with respect to in-
struction abstraction is Soot. The Jimple language from Soot is a bytecode-like
language using 3-address code instead of stack-based instructions, making it suit-
able for analysis and optimization [23].

6 Conclusions

Java bytecode is far from ideal for program analysis. Subroutines, a construct not
available in the Java language but only in Java bytecode, make data flow analy-
sis very complex. Eliminating subroutines is difficult because subroutines can be
nested, and they can overlap with exception handlers. In practice, inlining does not
increase program size much, while greatly simplifying data flow analysis. This is
especially valuable as subroutines are disappearing in modern compilers but still
have to be supported by virtual machines for backward compatibility.

Abstracting sets of similar instructions to a single instruction greatly reduces
the instruction set. Converting the stack-based representation to a register-based
one makes computational operands explicit and further reduces the instruction set.
Finally, eliminating certain bytecode-specific issues, such as wide instructions and
differences of 64-bit variables and stack elements, simplifies the code even further.
The resulting instruction set was successfully used in the JNuke framework for

static and dynamic analysis, which greatly benefits from the simplified bytecode
format.

